

# Exibição em Display de 7-Segmentos Com Entrada Analógica

Experimento 1

1<sup>st</sup> Iago Macarini Brito  
*Engenharia de Computação*  
*Universidade Tecnológica Federal do Paraná*  
Apucarana, Brasil  
iagomacarini@alunos.utfpr.edu.br

2<sup>nd</sup> Luis Henrique Ferracciu Pagotto Mendes  
*Engenharia de Computação*  
*Universidade Tecnológica Federal do Paraná*  
Apucarana, Brasil  
luis.mendes.2020@alunos.utfpr.edu.br

**Resumo**—Este relatório descreve o desenvolvimento de um sistema baseado em Arduino Mega 2560, capaz de converter sinais analógicos, provenientes de um potenciômetro, em valores digitais exibidos em dois displays de 7 segmentos. A montagem utiliza técnicas de multiplexação para otimizar o uso dos pinos digitais do microcontrolador, possibilitando a exibição de valores numérico de 00 a 99. O projeto demonstrou a eficiência da multiplexação e o correto funcionamento da leitura analógica, conversão digital e exibição visual dos dados.

**Palavras-Chave**—Arduino, multiplexação, potenciômetro, microcontrolador, sinal analógico

## I. INTRODUÇÃO

O processo de conversão de dados analógicos em representações digitais é uma prática comum no uso de sistemas embarcados, com muitas aplicações exigindo o uso de interfaces simples e de baixo custo. O Arduino Mega 2560 oferece um ambiente acessível para o desenvolvimento de protótipos que envolvem a entrada de sinais analógicos e sua conversão em representações digitais.

Este relatório apresenta a realização de um sistema que utiliza um potenciômetro como entrada analógica para controlar a exibição de valores numéricos em dois displays de 7 segmentos, variando de 00 a 99. O objetivo geral é ilustrar o ponto de encontro entre a aquisição de sinais analógicos e a exibição gráfica em displays digitais, utilizando técnicas de multiplexação para otimizar a eficiência na utilização dos pinos pelo microcontrolador.

## II. MATERIAIS E MÉTODOS

Para a realização deste experimento e verificação dos resultados, foram utilizados os materiais a seguir:

### A. Materiais Utilizados

- 1 Arduino Mega 2560;
- 2 Displays de 7-segmentos de catodo comum;
- 1 Protoboard;
- 1 Potenciômetro linear de 100k $\Omega$ ;
- 2 Resistores de 330 $\Omega$ ;
- Jumpers macho-macho;
- 1 Fonte de alimentação USB 5V.

### B. Montagem do Sistema

A implementação do sistema foi feita a partir do uso de dois displays de 7 segmentos para saída numérica (00-99), sendo controlados por um potenciômetro como entrada analógica. Inicialmente, os dois displays de 7 segmentos foram inseridos na protoboard e, em seguida, os segmentos correspondentes de cada display foram interligados (segmento "a" do primeiro display ao segmento "a" do segundo, e assim por diante até o segmento "g"), criando assim um barramento único para os componentes. Essa configuração permite acionar simultaneamente os segmentos correspondentes de ambos os displays por meio de um único conjunto de pinos digitais no Arduino.

Para que fosse possível o controle individual de cada display, resistores foram conectados aos cátodos comuns, os quais, por sua vez, foram ligados aos pinos digitais 40 e 42 do Arduino Mega 2560. Isso tornou viável o acendimento seletivo de cada display por meio do método de varredura temporal, ou multiplexação.

A explicação passo a passo de como os elementos do display interagem com o Arduino foi realizada da seguinte forma:

- Segmento a: Pino digital 22;
- Segmento b: Pino digital 24;
- Segmento c: Pino digital 26;
- Segmento d: Pino digital 28;
- Segmento e: Pino digital 30;
- Segmento f: Pino digital 32;
- Segmento g: Pino digital 34;

Para utilizar a aquisição de sinal analógico, o potenciômetro foi adicionado ao sistema conectando três de seus terminais: o terminal positivo ao pino de 5V do Arduino, o terminal central (cursor) ao pino A0 para leitura da tensão variável, e o terminal negativo ao pino de referência do microcontrolador (GND). O modo de operação do sistema se baseia na leitura constante do valor analógico entre 0 e 1023 fornecido pelo potenciômetro, convertendo-o em um valor entre 0 e 99, e posteriormente dividindo esse valor em dígitos separados para exibição gráfica nos displays. A rápida alternância entre os displays, que não é percebida pelo olho humano devido ao

efeito da persistência da visão, produz o efeito de exibição simultânea de ambos os dígitos.

### C. Código Implementado

```
const int segmentPins[7] =
    {22,24,26,28,30,32,34};
const int displayPins[2] = {42, 40};
const int potPin = A0;
const byte digitPatterns[10] = {
    B1111110, // Dígito 0
    B0110000, // Dígito 1
    B1101101, // Dígito 2
    B1111001, // Dígito 3
    B0110011, // Dígito 4
    B1011011, // Dígito 5
    B1011111, // Dígito 6
    B1110000, // Dígito 7
    B1111111, // Dígito 8
    B1111011 // Dígito 9
};

void setup() {
    for (int i = 0; i < 7; i++) {
        pinMode(segmentPins[i], OUTPUT);
    }
    for (int i = 0; i < 2; i++) {
        pinMode(displayPins[i], OUTPUT);
        digitalWrite(displayPins[i], HIGH);
    }
    Serial.begin(9600);
}

void loop() {
    int potValue = analogRead(potPin);
    Serial.print("Valor do potenciometro: ");
    Serial.print(potValue);

    int displayValue=map(potValue,0,1023,0,99);
    Serial.print(" | Valor ajustado (0-99): ");
    Serial.println(displayValue);

    int tens = displayValue / 10;
    int units = displayValue % 10;

    for (int refresh = 0; refresh < 20; refresh++) {
        displayDigit(0, tens);
        delay(5);
        displayDigit(1, units);
        delay(5);
    }
}

void displayDigit(int displayIndex,int digit)
{
    for (int i = 0; i < 2; i++) {
        digitalWrite(displayPins[i],HIGH);
    }
    for (int i = 0; i < 7; i++) {
        bool segmentOn=bitRead(digitPatterns[digit],6-i);
        digitalWrite(segmentPins[i],segmentOn?HIGH:LOW);
    }
    digitalWrite(displayPins[displayIndex],LOW);
}
```

Dado o código acima, o array *digitPatterns* contém os padrões binários para representar os números de 0 a 9 nos segmentos do display. Cada valor binário indica quais segmentos devem estar acesos para formar um número. Na função **setup**, todos os pinos dos segmentos e dos displays são configurados como saídas, e os displays são desligados inicialmente (colocados em estado *HIGH*, já que, nesse caso, provavelmente eles são do tipo catodo comum e são ativados com *LOW*). Também é iniciada a comunicação serial para mostrar no monitor serial os valores lidos no potenciômetro. No *loop*, o código lê o valor analógico do potenciômetro (de 0 a 1023) e o mapeia para um valor entre 0 e 99. Esse valor é dividido em dezenas e unidades, que são os dois números a serem exibidos. Como o display usa multiplexação (compartilha os mesmo pinos de segmento para os dois dígitos), é necessário acender um dígito de cada vez. Isso é feito dentro de um laço de repetição que alterna rapidamente entre o display das dezenas e o das unidades, com pequenos delays, criando a ilusão de que ambos estão acesos ao mesmo tempo. A função *displayDigit* é responsável por acender os segmentos corretos de acordo com o dígito que deve ser mostrado. Ela primeiro desliga ambos os displays, atualiza os segmentos de acordo com o dígito recebido, e por fim, liga apenas o display correspondente.

A seguir, a Figura 1 é a foto do experimento já montado, e em funcionamento, durante a fase de testes do código desenvolvido. A figura 2 é uma representação criada no software SimulIDE para facilitar o entendimento em relação à montagem do sistema.




Figura 1. Foto do Circuito em Funcionamento




Figura 2. Esquema Elétrico Representativo do Circuito

### III. CONCLUSÃO

Durante o desenvolvimento do sistema, foi possível entender e confirmar que a multiplexação é de grande utilidade para introduzir um ou mais displays de 7-segmentos. Com os testes realizados em bancada, foi verificada a implementação correta do mapeamento dos dados de entrada com o potenciômetro e a conversão desses no valores de 00 a 99.

Portanto, pode-se inferir que o circuito funcionou perfeitamente, a ponto que cada segmento dos displays acendeu corretamente e os valores ficaram visíveis com clareza, e a variação dos mesmos foram apresentados perfeitamente conforme variação no potenciômetro.